MLC Lab Visit - Lab 04 - Maple

Mth 355 (a.k.a. Mth 399) Jan 29, 2003 Maple 7

Bent E. Petersen
bent@alum.mit.edu
petersen@math.orst.edu

There are 6 problems below. Problem solutions are due Feb 5, 2003. Email your solutions to me as Maple worksheet attachments. Your worksheet must execute correctly for full credit.

This worksheet contains a few comments on some of Maple's recurrence relations support. If you find a serious error be sure to mention it in your solution report.

Linear Recurrence Equations - Examples

```
> restart;
```

Example. Fibonacci sequence

The Maple function rsolve() is used to solve linear recurrence relations. Here is an example of an order 2 recurrence relation with initial conditions. The solution is the Fibonacci sequence

```
> eqn1:=a(n)=a(n-1)+a(n-2);
```

$$eqn1 := \mathrm{a}(n) = \mathrm{a}(n-1) + \mathrm{a}(n-2)$$

```
> init1:=a(0)=0,a(1)=1;
```

$$init1 := \mathrm{a}(0) = 0,\ \mathrm{a}(1) = 1$$

To solve the recurrence equation we use rsolve(). Note how we specify for which variable to solve.

```
> soln1:=rsolve({eqn1,init1},a);
```

$$soln1 := \frac{\left(1-\frac{1}{5}\sqrt{5}\right)\left(2\frac{1}{-1+\sqrt{5}}\right)^n}{-1+\sqrt{5}} + \frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}$$

We can also use rsolve() to find the generating function for the solution. Note the single quotes surrounding genfunc in the following command.

```
> gen1:=rsolve({eqn1,init1},a,'genfunc'(z));
```

$$gen1 := -\frac{z}{-1 + z + z^2}$$

A very nice feature of Maple is that we can even define a procedure which will compute the a(n):

```
> fun1:=rsolve({eqn1,init1},a,'makeproc'):
```

Let's check it:

```
> fun1(3); fun1(4); fun1(5); fun1(6); fun1(7); fun1(8);
```

$$2$$
$$3$$
$$5$$
$$8$$
$$13$$
$$21$$

The fun1() procedure can be use to plot the sequence a(n)

```
> seq1:=seq([n,fun1(n)],n=0..15);
```

$seq1 := [0, 0], [1, 1], [2, 1], [3, 2], [4, 3], [5, 5], [6, 8], [7, 13], [8, 21], [9, 34], [10, 55], [11, 89], [12, 144], [13, 233], [14, 377], [15, 610]$

```
> plot([seq1],style=point,symbol=diamond,symbolsize=20,title="f
  un1 - Fibonacci sequence",color=blue);
```



If we omit the "style=point" directive Maple will produce the plot of the piecewise linear interpolation rather than the individual points:

```
> plot([seq1],title="fun1 - Fibonacci sequence" ,color=blue,
  thickness=2);
```




```
>
```

Example. Derangements

In class we obtained a recurrence equation for the number of derangements.

```
> eqn2:=d(n)=n*d(n-1)+(-1)^n;
```

$$eqn2 := d(n) = n\, d(n-1) + (-1)^n$$

```
> init2:=d(2)=1;
```

$$init2 := d(2) = 1$$

The initial condition just means the number of derangements of the set {1,2} is 1, which is clear.

```
> soln2:=rsolve({eqn2,init2},d);
```

$$soln2 := \mathbf{e}^{(-1)}\, \Gamma(n+1, -1)$$

Hmm. Maple's response is in terms of the Incomplete Gamma function - not a familiar object to most of us.

It is usually more convenient to have the initial condition at the origin, especially for computing the generating function in Maple, so let's see what the initial condition should be.

```
> fun2:=rsolve({eqn2,init2},d,'makeproc'):
> simplify(fun2(0));
```

$$1$$

This condition is a bit difficult to interpret. We can view it as saying there is one permutation of the empty set, the identity permutation, and it is a derangement since it has no fixed points.

Let's look at the exponential generating function. Maple can not compute it directly, so we compute the ordinary generating function for a(n)=d(n)/n! nstead:

```
> eqn2a:=a(n)=a(n-1)+(-1)^n/n!;
```

$$eqn2a := \mathrm{a}(n) = \mathrm{a}(n-1) + \frac{(-1)^n}{n!}$$

```
> init2a:=a(0)=1;
```

$$init2a := \mathrm{a}(0) = 1$$

```
> gen2:=rsolve({eqn2a,init2a},a,'genfunc'(z));
```

$$gen2 := -\frac{\left(\sum_{n=1}^{\infty} \frac{(-1)^n z^n}{n!}\right) + 1}{-1 + z}$$

```
> value(gen2): simplify(%);
```

$$-\frac{\mathbf{e}^{(-z)}}{-1 + z}$$

Well, we had to coax Maple a bit, but we got the right answer.

```
>
```

Example. A recurrence relation with complex roots

Here's a recurrence equation demonstrating damped oscillation:

```
> eqn3:=a(n)=a(n-1)-a(n-2)/2;
```

$$eqn3 := \mathrm{a}(n) = \mathrm{a}(n-1) - \frac{1}{2}\mathrm{a}(n-2)$$

```
> init3:=a(0)=1,a(1)=1;
```

$$init3 := \mathrm{a}(0) = 1, \mathrm{a}(1) = 1$$

```
> soln3:=rsolve({eqn3,init3},a);
```

$$soln3 := \left(\frac{1}{2} + \frac{1}{2} I\right)\left(\frac{1}{2} - \frac{1}{2} I\right)^n + \left(\frac{1}{2} - \frac{1}{2} I\right)\left(\frac{1}{2} + \frac{1}{2} I\right)^n$$

```
> gen3:=rsolve({eqn3,init3},a,'genfunc'(z));
```

$$gen3 := 2\,\frac{1}{2 - 2\,z + z^2}$$

```
> fun3:=rsolve({eqn3,init3},a,'makeproc'):
> seq3:=seq([n,fun3(n)],n=0..20);
```

$$seq3 := [0, 1], [1, 1], \left[2, \frac{1}{2}\right], [3, 0], \left[4, \frac{-1}{4}\right], \left[5, \frac{-1}{4}\right], \left[6, \frac{-1}{8}\right], [7, 0], \left[8, \frac{1}{16}\right], \left[9, \frac{1}{16}\right], \left[10, \frac{1}{32}\right], [11, 0], \left[12, \frac{-1}{64}\right], \left[13, \frac{-1}{64}\right], \left[14, \frac{-1}{128}\right], [15, 0], \left[16, \frac{1}{256}\right], \left[17, \frac{1}{256}\right], \left[18, \frac{1}{512}\right], [19, 0], \left[20, \frac{-1}{1024}\right]$$

```
> plot([seq3],style=point,symbol=box,symbolsize=20,color=blue,title="Damped vibration");
```




```
>
```

Linear Homogeneous Recurrence Equations of Finite Order

```
> restart;
```

For the linear homogeneous recurrence equation of order 3

```
> eqn:=a(n)=A[1]*a(n-1)+A[2]*a(n-2)+A[3]*a(n-3);
```

$$eqn := a(n) = A_1\, a(n-1) + A_2\, a(n-2) + A_3\, a(n-3)$$

the characteristic polynomial is given by

```
> charpoly:=subs(a(n-3)=1,a(n-2)=x,a(n-1)=x^2,a(n)=x^3,eqn);
```

$$charpoly := x^3 = A_1\, x^2 + A_2\, x + A_3$$

We can write a simple (not very robust) procedure to do this substitutuion for us:

```
> cp:=proc(recureqn,a,n,order,var)
>   local k,sq;
> sq:=seq(a(n-k)=var^(order-k),k=0..order);
> subs(sq,recureqn);
> end:
>
> cp(eqn,a,n,3,z);
```

$$z^3 = A_1\, z^2 + A_2\, z + A_3$$

```
>
```

Example 1

Let a(n) be the number of strings of 0's and 1's of length n which do not contain the bit pattern 11. If a such a string starts with 0 the remainder can be any string of length n-1 which does not contain 11. If on the other hand it starts with 1 then the second character must be 0 and after that we can have any string of length n-2 which does not contain 11. Thus (note we just have a shifted Fibonacci sequence)

```
> eqn4:=b(m)=b(m-1)+b(m-2);
```

$$eqn4 := \mathrm{b}(m) = \mathrm{b}(m-1) + \mathrm{b}(m-2)$$

```
> init4:=b(0)=1,b(1)=2;
```

$$init4 := \mathrm{b}(0) = 1,\ \mathrm{b}(1) = 2$$

```
> soln4:=rsolve({eqn4,init4},b(m));
```

$$soln4 := \frac{\left(-1 - \frac{1}{5}\sqrt{5}\right)\left(-2\,\frac{1}{1-\sqrt{5}}\right)^m}{1-\sqrt{5}} + \frac{\left(\frac{1}{5}\sqrt{5} - 1\right)\left(-2\,\frac{1}{1+\sqrt{5}}\right)^m}{1+\sqrt{5}}$$

Let's compute the characteristic polynomial and the characteristic roots to see how they match up to Maple's solution:

```
> cpoly:=cp(eqn4,b,m,2,z);
```

$$cpoly := z^2 = z + 1$$

```
> solve(cpoly,z);
```

$$\frac{1}{2}+\frac{1}{2}\sqrt{5}, \frac{1}{2}-\frac{1}{2}\sqrt{5}$$

```
> fun4:=rsolve({eqn4,init4},b,'makeproc'):
> for k from 1 to 8 do fun4(k); od;
```

2

3

5

8

13

21

34

55

```
>
```

Example 2

Here's another homogeneous linear recurrence relation

```
> eqn5:=a(n)=-a(n-2);
```

$$eqn5 := a(n) = -a(n-2)$$

```
> init5:=a(0)=2,a(1)=-1;
```

$$init5 := a(0) = 2, a(1) = -1$$

```
> soln5:=rsolve({eqn5,init5},a);
```

$$soln5 := \left(1-\frac{1}{2}I\right)(-I)^n + \left(1+\frac{1}{2}I\right)I^n$$

```
> fun5:=rsolve({eqn5,init5},a,'makeproc'):
> for k from 1 to 8 do fun5(k); od;
```

-1

-2

1

2

-1

-2

1

2

```
> gen5:=rsolve({eqn5,init5},a,'genfunc'(z));
```

$$gen5 := -\frac{-2+z}{1+z^2}$$

```
> seq5:=seq([n,fun5(n)],n=0..20);
```

seq5 := [0, 2], [1, -1], [2, -2], [3, 1], [4, 2], [5, -1], [6, -2], [7, 1], [8, 2], [9, -1], [10, -2], [11, 1], [12, 2], [13, -1], [14, -2], [15, 1], [16, 2], [17, -1], [18, -2], [19, 1], [20, 2]

```
> plot([seq5],style=point,symbol=circle,symbolsize=20,color=blue,title="Periodic example");
```



This plot is hard to interpret. Let's look at the piecewise linear plot instead:

```
> plot([seq5],color=blue,title="Periodic example");
```



The characteristic polynomial here is:

```
> cpoly5:=cp(eqn5,a,n,2,z);
```

$$cpoly5 := z^2 = -1$$

```
>
```

Linear Inhomogeneous Recurrence Equations of Finite Order

Example 1

```
> eqn6:=c(n)=c(n-1)+c(n-2)-n;
```

$$eqn6 := c(n) = c(n-1) + c(n-2) - n$$

```
> init6:=c(0)=0,c(1)=1;
```

$$init6 := c(0) = 0,\ c(1) = 1$$

```
> soln6:=rsolve({eqn6,init6},c);
```

$$soln6 := \frac{\left(\frac{1}{5}\sqrt{5} - 1\right)\left(-2\,\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}} + \frac{\left(-1 - \frac{1}{5}\sqrt{5}\right)\left(-2\,\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}} + n + 3$$

$$+\frac{(1+\sqrt{5})\left(-2\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}}+\frac{(1-\sqrt{5})\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}$$

```
> gen6:=rsolve({eqn6,init6},c,'genfunc'(z)); simplify(%);
```

$$gen6 := -\frac{\left(-2\frac{1}{1-z}-\frac{z}{(1-z)^2}\right)z^2+z}{-1+z+z^2}$$

$$-\frac{z(-4z+2z^2+1)}{(-1+z)^2(-1+z+z^2)}$$

```
> fun6:=rsolve({eqn6,init6},c,'makeproc');
```

fun6 := **proc**(n)

if 1 < nargs **then** 'procname'(args)

else expand((1 / 5*sqrt(5) − 1)*(−2*1 / (1 − sqrt(5)))^n / (1 − sqrt(5))
+ (−1 − 1 / 5*sqrt(5))*(−2*1 / (1 + sqrt(5)))^n / (1 + sqrt(5)) + n + 3
+ (−1 − sqrt(5))*(2*1 / (−1 + sqrt(5)))^n / (−1 + sqrt(5))
+ (1 − sqrt(5))*(−2*1 / (1 + sqrt(5)))^n / (1 + sqrt(5)))

end if

end proc

```
> seq6tmp:=seq([n,fun6(n)],n=0..2);
```

$$seq6tmp := \left[0, \frac{1}{5}\frac{\sqrt{5}}{1-\sqrt{5}}-\frac{1}{1-\sqrt{5}}-\frac{6}{5}\frac{\sqrt{5}}{1+\sqrt{5}}+3-\frac{1}{-1+\sqrt{5}}-\frac{\sqrt{5}}{-1+\sqrt{5}}\right],$$

$$\left[1, -\frac{2}{5}\frac{\sqrt{5}}{(1-\sqrt{5})^2}+\frac{2}{(1-\sqrt{5})^2}+\frac{\frac{12}{5}\sqrt{5}}{(1+\sqrt{5})^2}+4-\frac{2}{(-1+\sqrt{5})^2}-\frac{2\sqrt{5}}{(-1+\sqrt{5})^2}\right],$$

$$\left[2, \frac{4}{5}\frac{\sqrt{5}}{(1-\sqrt{5})^3}-\frac{4}{(1-\sqrt{5})^3}-\frac{24}{5}\frac{\sqrt{5}}{(1+\sqrt{5})^3}+5-\frac{4}{(-1+\sqrt{5})^3}-\frac{4\sqrt{5}}{(-1+\sqrt{5})^3}\right]$$

Unless we really need it, this expression is too complicated. Try 21 terms!

```
> seq6:=seq([n,round(0,evalf(fun6(n)))],n=0..20);
```

seq6 := [0, 0], [1, 1], [2, -1], [3, -3], [4, -8], [5, -16], [6, -30], [7, -53], [8, -91], [9, -153], [10, -254], [11, -418], [12, -684], [13, -1115], [14, -1813], [15, -2943], [16, -4772], [17, -7732], [18, -12522], [19, -20273], [20, -32815]

```
> plot([seq6],style=point,symbol=diamond,symbolsize=20,color=bl
  ue);
```




```
>
```

Example 2

```
> eqn7:=a(n)=3*a(n-1)+4*a(n-2)-12*a(n-3)-n^2;
```

$$eqn7 := a(n) = 3\,a(n-1) + 4\,a(n-2) - 12\,a(n-3) - n^2$$

```
> init7:=a(0)=2,a(1)=5,a(2)=13;
```

$$init7 := a(0) = 2,\ a(1) = 5,\ a(2) = 13$$

```
> soln7:=rsolve({eqn7,init7},a);
```

$$soln7 := -\frac{3}{10}\,3^n + \frac{11}{2}\,2^n + \frac{31}{270}\,(-2)^n - \frac{1}{3}\,(n+1)\left(\frac{1}{2}\,n+1\right) - \frac{8}{9}\,n - \frac{161}{54}$$

```
> gen7:=rsolve({eqn7,init7},a,'genfunc'(z)); simplify(%);
```

$$gen7 := \frac{\left(-9\,\frac{1}{1-z} - \frac{6\,z}{(1-z)^2} - z\left(\frac{1}{(1-z)^2} + \frac{2\,z}{(1-z)^3}\right)\right)z^3 + 2 - z - 10\,z^2}{1 - 3\,z - 4\,z^2 + 12\,z^3}$$

$$-\frac{16\,z^3 - 18\,z^4 + 6\,z^5 + 2 - 7\,z - z^2}{(-1+z)^3\,(1 - 3\,z - 4\,z^2 + 12\,z^3)}$$

```
> fun7:=rsolve({eqn7,init7},a,'makeproc'):
> for k from 1 to 10 do fun7(k); od;
```

5

13

26

$$54$$
$$85$$
$$123$$
$$12$$
$$-556$$
$$-3177$$
$$-11999$$

```
> unassign('k');
>
```

Example 3 Nonparallel lines

Consider n nonparallel lines in the plane and suppose no three of them meet in a point. The lines divide the plane into a(n) regions. If we already have n-1 lines and add one more line, it cuts each of the existing n-1 lines and so creates n new regions. (This statement may require some argument.) Thus

```
> eqn8:=a(n)=a(n-1)+n;
```

$$eqn8 := \mathrm{a}(n) = \mathrm{a}(n-1) + n$$

```
> init8:=a(0)=1;
```

$$init8 := \mathrm{a}(0) = 1$$

```
> soln8:=rsolve({eqn8,init8},a(n));
```

$$soln8 := (n+1)\left(\frac{1}{2}n+1\right) - n$$

```
>
```

Problems

Solve the following recursive equations. In each case find the generating function as well.

Problem 1

```
> a(n) = 2*a(n-1)-7; a(0)=8;
```

$$\mathrm{a}(n) = 2\,\mathrm{a}(n-1) - 7$$
$$\mathrm{a}(0) = 8$$

Problem 2

```
a(n) = 7*a(n-1)-11*a(n-2); a(0)=3,a(1)=-2;
```

$$\mathrm{a}(n) = 7\,\mathrm{a}(n-1) - 11\,\mathrm{a}(n-2)$$
$$\mathrm{a}(0) = 3,\ \mathrm{a}(1) = -2$$

Problem 3

```
> a(n) = a(n-1) + n^3; a(1)=1;
```

$$a(n) = a(n-1) + n^3$$
$$a(1) = 1$$

What does this problem have to do with

```
> Sum(k^3,k=1..n);
```

$$\sum_{k=1}^{n} k^3$$

What does it have to do with

```
> Sum(k,k=1..n)^2;
```

$$\left(\sum_{k=1}^{n} k\right)^2$$

Problem 4

```
> a(n) = a(n-1) + n^5; a(1)=1;
```

$$a(n) = a(n-1) + n^5$$
$$a(1) = 1$$

What does this problem have to do with

```
> Sum(k^5,k=1..n);
```

$$\sum_{k=1}^{n} k^5$$

Problem 5

```
> a(n) = 2*a(n-1)+11*a(n-2)-12*a(n-3)+5*2^n;
  a(0)=-1,a(1)=3,a(2)=5;
```

$$a(n) = 2\, a(n-1) + 11\, a(n-2) - 12\, a(n-3) + 5\; 2^n$$
$$a(0) = -1,\ a(1) = 3,\ a(2) = 5$$

```
>
>
>
>
```